



[조건부 서식]

1. 다음 중 아래의 [A1:E5] 영역에서 B열과 D열에만 배경색을 설정하기 위한 조건부 서식의 규칙으로 옳은 것은?

	A	B	C	D	E
1	자산코드	L47C	S22C	N71E	S34G
2	비품명	디스크	디스크	디스크	모니터
3	내용연수	4	3	3	5
4	경과연수	2	1	2	3
5	취득원가	550,000	66,000	132,000	33,000

- ① =MOD(COLUMNS(\$A1), 2)=1
- ② =MOD(COLUMNS(A\$1), 2)=0
- ③ =MOD(COLUMN(\$A1), 2)=0
- ④ =MOD(COLUMN(A\$1), 2)=0

[해설]

- 열 번호가 짝수인 열에 서식을 지정해야 하므로, COLUMN 함수를 이용하여 각 셀의 열 번호를 구합니다.
 - COLUMN(셀) : 주어진 '셀'의 열 번호를 구함
 - COLUMNS(셀 범위) : 주어진 '셀 범위'의 열 개수를 구함
- [A1:E5] 영역에 있는 각 셀을 모두 비교하여 셀 단위로 서식을 지정할 때는 =MOD(COLUMN(A1), 2)=0으로, 조건에 맞는 데이터가 있는 열 전체에 서식을 지정하려면 행 번호 앞에 \$를 붙여 =MOD(COLUMN(A\$1), 2)=0으로 입력합니다. 두 경우의 결과는 동일합니다.
- [A1:E5] 영역을 범위로 지정한 후 조건부 서식의 규칙을 =MOD(COLUMN(A\$1), 2)=0으로 지정하면 열별로 수식이 아래와 같이 변경되어 각 열을 비교합니다.
 - =MOD(COLUMN(A\$1), 2)=0
 - =MOD(COLUMN(B\$1), 2)=0
 - =MOD(COLUMN(C\$1), 2)=0
 - =MOD(COLUMN(D\$1), 2)=0
 - =MOD(COLUMN(E\$1), 2)=0

2. [A1:C3] 영역에 대해 조건부 서식의 규칙을 다음과 같이 설정할 경우 결과 화면으로 옳은 것은?

규칙 : =MOD(\$A1, 2) = MOD(A\$1, 2)

배경색 : 회색

①

	A	B	C
1	1	2	3
2	4	5	6
3	7	8	9

②

	A	B	C
1	1	2	3
2	4	5	6
3	7	8	9

③

	A	B	C
1	1	2	3
2	4	5	6
3	7	8	9

④

	A	B	C
1	1	2	3
2	4	5	6
3	7	8	9

[해설]

- =MOD(\$A1,2)를 입력했을 때의 각 셀의 결과는 다음과 같습니다.

	A	B	C
1	1	2	3
2	4	5	6
3	7	8	9

→

	A	B	C
1	1	1	1
2	0	0	0
3	1	1	1

- =MOD(A\$1,2)를 입력했을 때의 각 셀의 결과는 다음과 같습니다.

	A	B	C
1	1	2	3
2	4	5	6
3	7	8	9

→

	A	B	C
1	1	0	1
2	1	0	1
3	1	0	1

- 두 수식의 결과가 같은 셀에 서식이 적용되므로, 서식 적용 결과는 다음과 같습니다.

	A	B	C
1	1	2	3
2	4	5	6
3	7	8	9

[엑셀 함수]

3. 아래 워크시트에서 성취도[C2:C6]는 성취율[B2:B6]을 10%로 나눈 값만큼 표시한 것으로, 성취율이 70%를 초과하면 “■”를, 그 외는 “□”을 반복하여 표시하였다. 다음 중 이를 위한 수식으로 옳은 것은?

	A	B	C
1	성명	성취율	성취도
2	김양호	98%	■■■■■■■■■■
3	이숙경	75%	■■■■■■■■
4	양미진	65%	□□□□□□
5	이형도	85%	■■■■■■■■■
6	김인경	50%	□□□□□

- ① =REPLACE(QUOTIENT(B2, 10%), IF(B2>70%, “■”, “□”))
 ② =REPT(QUOTIENT(B2, 10%), IF(B2>70%, “■”, “□”))
 ③ =REPLACE(IF(B2>70%, “■”, “□”), QUOTIENT(B2, 10%))
 ④ =REPT(IF(B2>70%, “■”, “□”), QUOTIENT(B2, 10%))

[해설]

=REPT(IF(B2>70%, “■”, “□”), QUOTIENT(B2, 10%))

①

②

③


- ① IF(B2>70%, “■”, “□”) : [B2] 셀의 값 98%가 70%보다 크므로 ■를 반환합니다.
 ② QUOTIENT(B2, 10%) : [B2] 셀의 값 98%를 10%로 나눈 값 9를 반환합니다.
 ③ =REPT(①, ②) → =REPT(“■”, 9) : “■”를 9번 반복하여 표시합니다.

4. 다음 중 수식과 그 실행 결과값의 연결이 옳지 않은 것은?

- ① =DAYS(“2020-11-1”, “2020-10-1”) → 31
 ② =ROUNDDOWN(45.6789, 2) → 45.67
 ③ =SUMPRODUCT({1,2,3}, {5,6,7}) → 32
 ④ =SQRT(4)*(INT(-2)+POWER(2,3)) → 12

[해설]

- ① =DAYS(“2020-11-1”, “2020-10-1”) : ‘2020-11-1’에서 ‘2020-10-1’을 뺀 일수인 31을 반환합니다.
 ② =ROUNDDOWN(45.6789, 2) : 45.6789를 소수점 이하 둘째 자리로 자리 내림한 45.67을 반환합니다.
 ③ =SUMPRODUCT({1,2,3}, {5,6,7}) : 배열에서 대응하는 요소를 모두 곱하고 그 곱을 합한 (1×5) + (2×6) + (3×7) = 38을 반환합니다.



④ =SQRT(4)*(INT(-2)+POWER(2, 3))

①

②

③

- ① SQRT(4) : 4의 양의 제곱근인 2를 반환합니다.
 ② INT(-2) : -2보다 크지 않은 정수인 -2를 반환합니다.
 ③ POWER(2, 3) : 2를 3번 곱한 8을 반환합니다.
 ∴ =①*(②+③) = 2*(-2+8) = 12

5. 다음 중 아래의 워크시트에서 '윤정희' 사원의 근속년수를 오늘 날짜를 기준으로 구하고자 할 때, [E11] 셀에 입력할 수식으로 옳은 것은?

	A	B	C	D	E
1					
2					
3		부서	이름	입사일	연봉
4		영업부	김나미	2020-03-01	3,000만 원
5		총무부	김보라	2019-03-02	3,500만 원
6		총무부	이지선	2016-03-02	3,200만 원
7		영업부	윤정희	2018-03-02	2,000만 원
8		총무부	임형석	2020-11-26	1,800만 원
9		총무부	서민규	2019-10-08	2,200만 원
10		총무부	김상희	2015-06-17	1,500만 원
11		이름	윤정희	근속년	3

- ① =YEAR(TODAY())-YEAR(VLOOKUP(C11, B4:E10, 2, 0))
 ② =YEAR(TODAY())-YEAR(HLOOKUP(C11, B4:E10, 2, 0))
 ③ =YEAR(TODAY())-YEAR(VLOOKUP(C11, C4:E10, 2, 0))
 ④ =YEAR(TODAY())-YEAR(HLOOKUP(C11, C4:E10, 2, 0))

[해설]

=YEAR(TODAY()) - YEAR(VLOOKUP(C11, C4:E10, 2, 0))

- ① TODAY() : 오늘의 날짜를 구합니다(오늘 날짜를 2023-5-1로 가정함).
 ② YEAR(①) : 오늘의 날짜에서 년도만을 표시합니다(2023).
 ③ VLOOKUP(C11, C4:E10, 2, 0) : [C4:E10] 영역의 첫 번째 열에서 윤정희(C11)와 정확히 일치하는 값(옵션이 '0'이므로)을 찾은 후 이 값이 있는 행에서 지정된 열(2) 위치에 있는 데이터를 입력합니다(2018-03-02).
 ④ YEAR(③) : '2018-03-02'에서 년도만 표시하므로 2018이 됩니다.
 ※ 2023-2018은 5입니다.

6. 다음 중 10,000,000원을 2년간 연 5.5%의 이자율로 대출할 때, 매월 말 상환해야 할 불입액을 구하기 위한 수식으로 옳은 것은?

- ① =PMT(5.5%/12, 12*2, -10000000)
 ② =PMT(5.5%, 12*2, -10000000)
 ③ =PMT(5.5%, 12*2, -10000000,0,1)
 ④ =PMT(5.5%/12, 12*2, -10000000,0,1)

[해설]

=PMT(5.5%/12, 2*12, -10000000)

- PMT 함수는 PMT(이자, 기간, 현재 가치, 미래 가치, 납입 시점)과 같이 인수를 지정하며, 각각의 인수는 다음과 같이 적용됩니다.
- 이자 : 이율이 연 단위이므로 12로 나누면 5.5%/12입니다.
- 기간 : 기간이 년 단위이므로 12를 곱하면 2*12입니다.
- 현재가치 : 대출금을 현재 받았으므로 현재 가치이고, 결과가 양수로 나오도록 음수로 입력하면 -10000000입니다.
- 미래가치 : 0이므로 생략합니다.
- 납입시점 : 매월 말이므로 생략합니다.

7. 아래 시트에서 판매금액이 3,000,000 이상인 제품의 개수를 구하는 배열 수식으로 맞는 것은? (단 판매금액은 '판매단가 × 수량'이다.)

	A	B	C
1	제품명	판매단가	수량
2	컴퓨터	750,000	5
3	노트북	1,200,000	2
4	모니터	540,000	3
5	프린터	653,000	7

- ① {=SUM((B2:C2*B5:C5>=3000000)*1)}
 ② {=COUNT((B2:C2*B5:C5>=3000000)*1)}
 ③ {=SUM((B2:B5*C2:C5>=3000000)*1)}
 ④ {=COUNT((B2:B5*C2:C5>=3000000)*1)}

[해설]

조건이 하나일 때 배열 수식을 이용하여 개수를 구하는 방법은 다음의 3가지 방법이 있습니다.

- 방법1 : {=SUM((조건1) * (조건2))}
- 방법2 : {=SUM(IF(조건1, IF(조건2, 1)))}
- 방법3 : {=COUNT(IF(조건1, IF(조건2, 1)))}

1. 조건과 범위 찾기

- 조건 : 판매금액이 3,000,000 이상이란 조건은, 비교 대상이 될 판매금액과 비교할 기준이 되는 3,000,000을 ">="로 연결하여 적어주면 됩니다(판매금액>=3000000).
- 판매금액 : 판매단가와 같은 위치에 대응하는 수량을 모두 곱하여 동시에 결과를 계산하려면 'B2:B5*C2:C5'로 입력합니다.

∴ B2:B5*C2:C5>=3000000

2. 위의 조건을 개수 구하기 배열 수식에 대입하면 다음과 같습니다.

- 방법1 : {=SUM((B2:B5*C2:C5>=3000000)*1)}
- 방법2 : {=SUM(IF(B2:B5*C2:C5>=3000000, 1))}
- 방법3 : {=COUNT(IF(B2:B5*C2:C5>=3000000, 1))}

8. 다음 중 아래 시트에서 <변경 전> 내용을 <변경 후>와 같이 변경하는 수식으로 옳은 것은?

	A
1	<변경 전>
2	서울시 도봉구 459 남위 36 북위 36
3	<변경 후>
4	서울시 도봉구 459 남위 136 북위 36

- ① =SUBSTITUTE(A2, "136", "36", 1)
 ② =SUBSTITUTE(A2, "136", "36", 2)
 ③ =SUBSTITUTE(A2, "36", "136", 1)
 ④ =SUBSTITUTE(A2, "36", "136", 2)

[해설]

- SUBSTITUTE(텍스트, 인수1, 인수2, n번째) 함수는 '텍스트'에서 '인수1'을 찾아, n번째에 있는 '인수1'을 '인수2'로 변경합니다.
- <변경 전>과 <변경 후>를 비교하면 [A2] 셀에 입력된 텍스트 중 첫 번째에 있는 "36"을 "136"으로 변경하였으므로 =SUBSTITUTE(A2, "36", "136", 1)로 지정하면 됩니다.

9. 다음 중 수식의 결과가 옳지 않은 것은?

- ① =FIXED(3456.789, 1, FALSE) → 3,456.8
- ② =EOMONTH(DATE(2015, 2, 25), 1) → 2015-03-31
- ③ =CHOOSE(ROW(A3:A6), “동”, “서”, “남”, 2015) → 남
- ④ =REPLACE(“February”, SEARCH(“U”, “Seoul-Unesco”), 5, “ ”) → Febru

[해설]

- ① =FIXED(3456.789, 1, FALSE) : FIXED(인수, 자릿수, 논리값)는 인수를 반올림하여 지정된 자릿수까지 텍스트로 표시합니다. 논리값이 FALSE이거나 생략되면 텍스트에 십표가 포함되므로 3456.789를 소수점 첫째 자리(1)로 반올림한 값 3,456.8이 표시됩니다.
- ② =EOMONTH(DATE(2015, 2, 25), 1) : EOMONTH(날짜, 월수)는 지정한 날짜를 기준으로 몇 개월 이전 또는 이후 달의 마지막 날짜의 일련번호를 구하는 함수입니다. 2015년 2월 25일을 기준으로 1개월 이후 달의 마지막 날짜는 2015년 3월 31일입니다.

- ③ =CHOOSE(ROW(A3:A6), “동”, “서”, “남”, 2015)

①

②

- ① ROW(A3:A6) : ROW(셀)는 주어진 셀의 행 번호를 구하는 함수입니다. ‘ROW(A3:A6)’과 같이 ROW 함수의 인수를 특정 셀이 아닌 범위를 지정하면 범위의 첫 번째 셀인 ‘A3’ 셀의 행 번호를 구합니다. A3 셀의 행 번호인 3을 반환합니다.

- ② =CHOOSE(①, “동”, “서”, “남”, 2015) → =CHOOSE(3, “동”, “서”, “남”, 2015) : CHOOSE(인수, 첫 번째, 두 번째, …)는 인수가 1이면 첫 번째를, 인수가 2이면 두 번째를 선택하는 함수입니다. “동”, “서”, “남”, 2015 중 세 번째에 있는 값인 “남”이 표시됩니다.

- ④ =REPLACE(“February”, SEARCH(“U”, “Seoul-Unesco”), 5, “ ”)

①

②

- ① SEARCH(“U”, “Seoul-Unesco”) : SEARCH(텍스트1, 텍스트2, 시작 위치)는 ‘텍스트2’에서 ‘시작 위치’부터 ‘텍스트1’을 찾아 위치를 표시하는 함수로 영문 대·소문자를 구분하지 않습니다. “Seoul-Unesco”에서 “U”를 찾아 위치를 표시하면 4입니다.

※ 시작 위치를 생략하면 첫 번째 글자부터 찾아 표시합니다.

- ② =REPLACE(“February”, ①, 5, “ ”) → =REPLACE(“February”, 4, 5, “ ”) : REPLACE(텍스트1, 시작 위치, 개수, 텍스트2)는 ‘텍스트1’의 ‘시작 위치’에서 ‘개수’로 지정된 문자를 ‘텍스트2’로 변경하는 함수입니다. “February”에서 네 번째 글자부터 다섯 글자를 빈칸으로 변경하면 “Feb”입니다.

10. 아래 워크시트에서 자격증 응시자에 대한 과목별 점수의 합계를 배열 수식으로 구하였다. 다음 중 [C10] 셀에 입력된 배열 수식으로 옳은 것은?

	A	B	C
1	응시자	과목	점수
2	김영호	1과목	60
3		2과목	85
4	강미진	1과목	90
5		2과목	75
6	최수영	1과목	80
7		2과목	95
8			
9		과목	합계
10		1과목	230
11		2과목	255
12			

- ① {=SUM(IF(\$B\$2:\$B\$7=B10, \$C\$2:\$C\$7))}
 ② {=SUM(IF(MOD(ROW(\$C\$2:\$C\$7), 2)=1, \$C\$2:\$C\$7))}
 ③ {=SUM(IF(C2:C7, B2:B7=B10))}
 ④ {=SUM(IF(MOD(ROWS(C2:C7), 2)=0, C2:C7))}

[해설]

- 합계를 구하는 배열 수식은 다음의 두 가지 식이 있으며, 조건의 개수에 따라 조건을 지정하는 부분만 늘어납니다.

- 방법 1 : {=SUM((조건) * 합계를_구할_범위) }
- 방법 2 : {=SUM(IF((조건), 합계를_구할_범위)) }

1. 조건과 범위 찾기

- 조건 : 과목별 점수란 조건은, 비교 대상이 될 지점 범위 [B2:B7]와 비교할 기준이 되는 “1과목”이 들어있는 [B10] 셀을 “=”으로 연결하여 적어주면 됩니다(B2:B7=B10).
- 합계를_구할_범위 : 점수이므로 C2:C7이 됩니다.

2. 위의 조건과 범위를 합계 구하기 배열 수식에 대입하면 다음과 같습니다.

- 방법 1 : =SUM((B2:B7=B10) * C2:C7)
- 방법 2 : =SUM(IF(B2:B7=B10, C2:C7))

- 이 문제는 여러 결과값을 구하는 수식으로, 범위를 절대 참조로 지정하여 작성해야 합니다.
- ‘방법2’를 절대 참조로 지정하여 입력한 후 [Ctrl]+[Shift]+[Enter]를 누르면 중괄호({ })가 표시되므로 {=SUM(IF(\$B\$2:\$B\$7=B10, \$C\$2:\$C\$7))}로 표시됩니다.

11. 다음과 같은 시트에서 [A8] 셀에 아래의 수식을 입력했을 때 계산 결과로 올바른 것은?

=COUNT(OFFSET(D6, -5, -3, 2, 2))

	A	B	C	D
1	성명	중간	기말	합계
2	김나희	100	80	180
3	김근석	90	95	185
4	배정희	80	63	143
5	탁지연	95	74	169
6	한정희	55	65	120

- ① 4
 ② 1
 ③ 120
 ④ 74

[해설]

=COUNT(OFFSET(D6, -5, -3, 2, 2))

①

②

① OFFSET(D6, -5, -3, 2, 2) : [D6] 셀을 기준으로 -5행, -3열 떨어진 셀 주소([A1])를 찾습니다. 이 주소를 기준으로 2행, 2열의 범위(A1:B2)를 지정합니다.

※ OFFSET(범위, 행, 열, 높이, 너비) 함수에서 행과 열로 지정한 인수가 음수(-)일 경우에는 선택한 범위에서 위쪽(행) 또는 왼쪽(열)으로 이동합니다.

② =COUNT(①) → =COUNT(A1:B2) : [A1:B2] 영역에서 수치 데이터(B2)의 개수를 구합니다(1).

12. 다음 워크시트에서 [G3:G6] 영역에 월요일부터 금요일까지 모두 출석(√)하면 “우수”, 그렇지 않으면 빈칸을 표시하려고 할 때 옳은 수식은?

	A	B	C	D	E	F	G
1	출석						
2	이름	월	화	수	목	금	비고
3	홍길동	√	√	√	√	√	
4	이대한	√	√		√		
5	김우리	√	√	√	√	√	
6	이석경	√		√	√	√	

- ① =IF(COUNT(B3:F3)=5, “우수”, “”)
 ② =IF(COUNTA(B3:F3)=5, “우수”, “”)
 ③ =IF(NOT(COUNTBLANK(B3:F3)=5), “우수”, “”)
 ④ =IF(COUNTIF(B3:F3, “”) =5, “”, “우수”)

[해설]

=IF(COUNTA(B3:F3)=5, “우수”, “”)

①

②

① COUNTA(B3:F3) : [B3:F3] 영역의 셀 개수인 5를 반환합니다.

② =IF(①=5, “우수”, “”) → =IF(5=5, “우수”, “”) : 조건이 참이므로 우수를 반환합니다.

13. 아래 워크시트에서 매출액[B3:B9]을 이용하여 매출 구간별 빈도수를 [F3:F6] 영역에 계산한 후 그 값만큼 “★”를 반복하여 표시하고자 한다. 다음 중 이를 위한 수식으로 옳은 것은?

	A	B	C	D	E	F
1						
2		매출액		매출구간		빈도수
3		75		0	50	★
4		93		51	100	★★
5		130		101	200	★★★
6		32		201	300	★
7		123				
8		257				
9		169				

- ① =REPT(“★”, FREQUENCY(E3:E6, B3:B9))
 ② =REPT(“★”, FREQUENCY(B3:B9, E3:E6))
 ③ {=REPT(“★”, FREQUENCY(E3:E6, B3:B9))}
 ④ {=REPT(“★”, FREQUENCY(B3:B9, E3:E6))}

[해설]

{=REPT(“★”, FREQUENCY(B3:B9, E3:E6))}

①

②

① FREQUENCY(B3:B9, E3:E6) : [B3:B9] 영역의 데이터를 대상으로 [E3:E6] 영역의 구간별 빈도수를 계산합니다.

② =REPT(“★”, ①) : “★”를 ①의 결과값만큼 반복하여 표시합니다.

※ FREQUENCY 함수는 결과가 여러 개의 값을 갖는 배열로 반환되므로 배열 수식으로 작성해야 합니다.

• 결과가 계산될 [F3:F6] 영역을 블록으로 지정하고 =REPT(“★”, FREQUENCY(B3:B9, E3:E6))을 입력한 후 [Ctrl]+[Shift]+[Enter]를 누르면 수식 앞뒤에 중괄호({ })가 자동으로 입력되어 {=REPT(“★”, FREQUENCY(B3:B9, E3:E6))}과 같이 표시됩니다.

14. 아래 그림과 같이 워크시트에 배열 상수 형태로 배열 수식이 입력되어 있을 때, [A5] 셀에서 수식 =SUM(A1, B2)를 실행하였다. 다음 중 그 결과로 옳은 것은?

	A	B	C
1	= {1,3,5;2,4,6}	= {1,3,5;2,4,6}	= {1,3,5;2,4,6}
2	= {1,3,5;2,4,6}	= {1,3,5;2,4,6}	= {1,3,5;2,4,6}

- ① 3
 ② 5
 ③ 6
 ④ 7

[해설]

• 배열 상수를 입력할 때 열의 구분은 쉼표(,)로, 행의 구분은 세미콜론(;)으로 합니다.

• [A1:C2] 영역을 블록으로 지정한 후 = {1,3,5;2,4,6}을 입력하고 [Ctrl]+[Shift]+[Enter]를 누르면 다음과 같이 입력됩니다.

	A	B	C
1	1	3	5
2	2	4	6

※ 문제에 제시된 그림은 이 상태에서 [Ctrl]+[~]를 눌러 값이 아닌 수식을 화면에 표시한 것입니다.

15. 아래의 시트에서 [A8] 셀에 '=INDEX(A1:C6, MATCH(LARGE(C2: C6,3), C1:C6, 0), 2)' 수식을 입력했을 때의 계산 결과로 올바른 것은?

	A	B	C
1	코너	담당	판매금액
2	잡화	김남희	5,122,000
3	식료품	남궁민	450,000
4	잡화	이수진	5,328,000
5	식료품	서수남	6,544,000
6	식료품	김정미	6,024,500

- ① 남궁민
- ② 이수진
- ③ 서수남
- ④ 김정미

[해설]

=INDEX(A1:C6, MATCH(LARGE(C2:C6, 3), C1:C6, 0), 2)

- ① LARGE(C2:C6, 3) : [C2:C6] 영역에서 3번째로 큰 값을 구하면 결과는 5328000입니다.
- ② MATCH(①, C1:C6, 0) → MATCH(5328000, C1:C6, 0) : [C1:C6] 영역에서 5328000과 정확히 일치하는 값을 찾은 후 상대 위치를 표시하면 결과는 4입니다.
- MATCH(찾을 값, 범위, 옵션) 함수에서 '옵션'을 0으로 지정하면 '찾을 값'과 정확히 일치하는 값을 찾습니다.
- ③ =INDEX(A1:C6, ②, 2) → =INDEX(A1:C6, 4, 2) : [A1:C6] 영역에서 4행 2열 즉, [B4] 셀의 값 이수진을 반환합니다.

16. 다음 그림과 같이 '성'과 '이름'을 합쳐서 '성명'으로 표시하고자 할 때, [C2] 셀에 들어갈 알맞은 수식은?

	A	B	C
1	성	이름	성명
2	이	덕환	이덕환
3	안	치연	안치연
4	강	청기	강청기
5	연	구현	연구현

- ① =PROPER(A2, B2)
- ② =REPLACE(A2, B2)
- ③ =CONCAT(A2, B2)
- ④ =TEXT(A2, B2)

[해설]

- PROPER(텍스트) : '텍스트'의 첫 글자만 대문자로 표시함
- REPLACE(텍스트1, 시작 위치, 개수, 텍스트2) : '텍스트1'의 '시작 위치'에서 '개수'로 지정된 문자를 '텍스트2'로 변경함
- CONCAT(텍스트1, 텍스트2, ...) : 인수로 주어진 텍스트들을 연결하여 1개의 문자열로 합침
- TEXT(인수, 형식) : '인수'를 지정한 '형식'의 텍스트로 바꿈

17. 다음 중 수식의 결과가 옳지 않은 것은?

	A
1	바나나
2	사과
3	오렌지
4	PEAR
5	3.14659

① =FIXED(A5, , FALSE) → 3.14

② =REPT("◆", LEN(A4)) → ◆◆◆◆

③ {=TEXT(SUM(IF(ISTEXT(A1:A5), 1, 0)), "과일의 수는 0개")} → 과일의 수는 4개

④ =REPLACE(A3, 2, 2, "가피나무") → 오가피나무

[해설]

① =FIXED(A5, , FALSE) : FIXED 함수의 '자릿수'를 생략하면 2로 지정되고, '논리값'을 FALSE 또는 생략하면 쉼표를 포함하므로 3.14659를 소수점 둘째 자리로 반올림한 3.15를 반환합니다.

※ [A5] 셀의 값 3.14659는 정수 부분이 한 자리이므로 쉼표, 즉 천 단위 구분 기호는 표시되지 않습니다. 예를 들어 [A5] 셀의 값이 1234.14659라면 1,234.15로 표시됩니다.

② =REPT("◆", LEN(A4)) : "◆"를 [A4] 셀(PEAR)의 글자 수만큼 4번 반복한 ◆◆◆◆를 반환합니다.

③ {=TEXT(SUM(IF(ISTEXT(A1:A5), 1, 0)), "과일의 수는 0개")}

① SUM(IF(ISTEXT(A1:A5), 1, 0)) : 조건에 만족하는 셀의 개수를 구하는 배열 수식으로, [A1:A5] 영역에서 인수가 텍스트인 셀의 개수인 4를 반환합니다.

② {=TEXT(①, "과일의 수는 0개")} → {=TEXT(4, "과일의 수는 0개")} : 4를 "과일의 수는 0개" 형식으로 표시한 과일의 수는 4개를 반환합니다.

④ =REPLACE(A3, 2, 2, "가피나무") : [A3] 셀의 값 "오렌지"의 2번째부터 2글자를 "가피나무"로 변경한 오가피나무를 반환합니다.

18. 다음 중 수식의 결과가 나머지 셋과 다른 것은?

	A	B
1	제품명	개수
2	건조기	1
3	김치냉장고	#N/A
4	냉장고	3
5	세탁기	TRUE
6	식기세척기	5

- ① =CHOOSE(ROWS(A2:B6), A2, A3, A4, A5, A6)
 ② =CHOOSE(N(B5), A2, A3, A4, A5, A6)
 ③ =CHOOSE(CELL("contents", B2), A2, A3, A4, A5, A6)
 ④ =CHOOSE(TYPE(B4), A2, A3, A4, A5, A6)

[해설]

- ① =CHOOSE(ROWS(A2:B6), A2, A3, A4, A5, A6)

①

②

- ① ROWS(A2:B6) : [A2:B6] 영역의 행의 수인 5를 반환합니다.
 ② =CHOOSE(①, A2, A3, A4, A5, A6) → =CHOOSE(5, A2, A3, A4, A5, A6) : 다섯 번째에 있는 [A6] 셀의 값인 식기세척기를 반환합니다.
 ③ =CHOOSE(N(B5), A2, A3, A4, A5, A6) : N(B5)는 [B5] 셀의 값 'TRUE'의 숫자값 1을 반환하므로, 결과는 첫 번째에 있는 [A2] 셀의 값인 건조기를 반환합니다.
 ④ =CHOOSE(CELL("contents", B2), A2, A3, A4, A5, A6) : CELL("contents", B2)에서 'contents'는 셀의 값을 의미하므로 [B2] 셀의 값인 1을 반환하므로 결과는 첫 번째에 있는 [A2] 셀의 값인 건조기를 반환합니다.
 ⑤ =CHOOSE(TYPE(B4), A2, A3, A4, A5, A6) : TYPE(B4)는 [B4] 셀에 입력된 값이 숫자이므로 1을 반환하므로 결과는 첫 번째에 있는 [A2] 셀의 값인 건조기를 반환합니다.

19. 다음 중 아래의 워크시트에서 수식의 결과로 “부사장”을 출력하지 않는 것은?

	A	B	C	D
1	사원번호	성명	직함	생년월일
2	101	구민정	영업 과장	1980-12-08
3	102	강수영	부사장	1965-02-19
4	103	김진수	영업 사원	1991-08-03
5	104	박용만	영업 사원	1990-09-19
6	105	이순신	영업 부장	1971-09-20

① =CHOOSE(CELL(“row”, B3), C2, C3, C4, C5, C6)

② =CHOOSE(TYPE(B4), C2, C3, C4, C5, C6)

③ =OFFSET(A1:A6, 2, 2, 1, 1)

④ =INDEX(A2:D6,MATCH(A3, A2:A6, 0), 3)

[해설]

① =CHOOSE(CELL(“row”, B3), C2, C3, C4, C5, C6)

- ① CELL(“row”,B3) : ‘row’는 행 번호를 의미하므로 CELL 함수는 [B3] 셀의 행 번호인 3을 반환합니다.
- ② =CHOOSE(①, C2, C3, C4, C5, C6) → =CHOOSE(3, C2, C3, C4, C5, C6) : 세 번째에 있는 [C4] 셀의 값을 표시하므로 결과는 영업 사원입니다.

② =CHOOSE(TYPE(B4), C2, C3, C4, C5, C6)

- ① TYPE(B4) : [B4] 셀에 입력된 값이 텍스트이므로 결과는 2입니다.
- ② =CHOOSE(①, C2, C3, C4, C5, C6) → =CHOOSE(2, C2, C3, C4, C5, C6) : 두 번째에 있는 [C3] 셀의 값을 표시하므로 결과는 부사장입니다.

③ =OFFSET(A1:A6, 2, 2, 1, 1) : [A1:A6] 영역의 첫 번째 셀인 [A1] 셀을 기준으로 2행 2열 떨어진 셀 주소([C3])를 찾습니다. 이 주소를 기준으로 1행 1열인 셀 즉, [C3] 셀의 값을 표시하므로 결과는 부사장입니다.

④ =INDEX(A2:D6, MATCH(A3, A2:A6, 0), 3)

- ① MATCH(A3, A2:A6, 0) : [A2:A6] 영역에서 [A3] 셀의 값과 동일한 값을 찾은 후 상대 위치를 표시하면 결과는 2입니다.
- ② =INDEX(A2:D6, ①, 3) → =INDEX(A2:D6, 2, 3) : [A2:D6] 영역에서 2행 3열 즉, [C3] 셀의 값을 표시하므로 결과는 부사장입니다.

20. 아래 워크시트에서 순위[G2:G10]는 총점을 기준으로 구하되 동점자에 대해서는 국어를 기준으로 순위를 구하였다. 다음 중 [G2] 셀에 입력된 수식으로 옳은 것은?

	A	B	C	D	E	F	G
1	성명	국어	수학	영어	사회	총점	순위
2	홍길동	92	50	30	10	182	1
3	한민국	80	50	20	30	180	3
4	이대한	90	40	20	30	180	2
5	이나래	70	50	30	30	180	4
6	마상욱	80	50	30	10	170	7
7	박정인	90	40	20	20	170	6
8	사수영	70	40	30	30	170	8
9	고소영	85	40	30	20	175	5
10	장영수	70	50	10	5	135	9

- ① {=RANK(\$F2, \$F\$2:\$F\$10)+RANK(\$B\$2, \$B\$2:\$B\$10)}
- ② {=RANK(\$B\$2, \$B\$2:\$B\$10)*RANK(\$F2, \$F\$2:\$F\$10)}
- ③ {=RANK(\$F2, \$F\$2:\$F\$10)+SUM((\$F\$2:\$F\$10=\$F2)*(\$B\$2:\$B\$10>\$B2))}
- ④ {=SUM((\$F\$2:\$F\$10=\$F2)*(\$B\$2:\$B\$10>\$B2))*RANK(\$F2, \$F\$2:\$F\$10)}

[해설]

‘총점’을 기준으로 순위를 구하되, 동점자에 대해 ‘국어’를 기준으로 순위를 구하는 방법은 다음과 같습니다.

- ① ‘총점’을 기준으로 순위를 구합니다.
- ② ‘총점’이 동일한 경우에는 동점자들의 ‘국어’ 점수를 비교하여 동점자 각각에 대해 자신의 ‘국어’ 점수보다 높은 점수의 개수를 자신의 ‘총점’ 순위에 더합니다.

	A	B	C	D	E	F	G	H
1	성명	국어	수학	영어	사회	총점	총점순위	순위
2	홍길동	92	50	30	10	182	1	1
3	한민국	80	50	20	30	180	2	3
4	이대한	90	40	20	30	180	2	2
5	이나래	70	50	30	30	180	2	4
6	마상욱	80	50	30	10	170	6	7
7	박정인	90	40	20	20	170	6	6
8	사수영	70	40	30	30	170	6	8
9	고소영	85	40	30	20	175	5	5
10	장영수	70	50	10	5	135	9	9

예) ‘한민국’의 순위는 ‘총점’을 기준으로 2위이지만, ‘총점’이 같은 동점자 3명 중 자신의 ‘국어’ 점수 80점보다 높은 사람이 1명이므로 최종 순위는 2위에 1을 더한 3위가 됩니다.

=RANK(\$F2, \$F\$2:\$F\$10)+SUM((\$F\$2:\$F\$10=\$F2)*(\$B\$2:\$B\$10>\$B2))

①

②

- ① RANK(\$F2, \$F\$2:\$F\$10) : [F2:F10] 영역에서 [F2] 셀의 순위를 구합니다. 여러 셀에 결과를 구해야 하므로 범위는 절대 참조로 지정해야 하지만, [F2] 셀의 경우는 F3, F4 등으로 변경되어야 하므로 F2 또는 \$F2로 지정하면 됩니다.

- ② SUM((\$F\$2:\$F\$10=\$F2)*(\$B\$2:\$B\$10>\$B2))

• 조건이 두 개일 때 배열 수식을 이용하여 개수를 구하는 방법은 다음의 3가지 방법이 있습니다.

- 방법1 : {=SUM((조건1) * (조건2))}
- 방법2 : {=SUM(IF(조건1, IF(조건2, 1)))}
- 방법3 : {=COUNT(IF(조건1, IF(조건2, 1)))}

1. 조건과 범위 찾기

- 조건1 : 총점이 동점인지를 비교해야 합니다. 비교 대상이 될 총점이 있는 범위(F2:F10)와 비교할 기준이 되는 [F2] 셀을 “=”으로 연결하여 적어주면 됩니다(F2:F10=F2).
- 조건2 : 동점자 중 국어 점수가 기준이 되는 국어 점수보다 높은 점수를 찾아야 합니다. 비교

대상이 될 국어가 있는 범위(B2:B10)와 비교할 기준이 되는 [B2] 셀을 ">"로 연결하여 적어주면 됩니다(B2:B10>B2).

2. 위의 조건을 개수 구하기 배열 수식에 대입하면 다음과 같습니다.

- 방법1 : {=SUM((F2:F10=F2)*(B2:B10>B2))}
- 방법2 : {=SUM(IF(F2:F10=F2, IF(B2:B10>B2, 1)))}
- 방법3 : {=COUNT(IF(F2:F10=F2, IF(B2:B10 >B2, 1)))}

- 여러 셀에 결과를 구해야 하므로 범위는 절대 참조로 지정해야 하고, [F2]와 [B2] 셀은 [F2] 또는 [F2], [B2] 또는 [\$B2]로 지정하면 됩니다.
- '방법1'을 적용하여 수식을 입력한 후 [Ctrl]+[Shift]+[Enter]를 누르면 중괄호({ })가 자동으로 붙여져 {=RANK(\$F2, \$F\$2:\$F\$10)+SUM((\$F\$2:\$F\$10=\$F2)*(\$B\$2:\$B\$10>\$B2))}로 표시됩니다.

21. 아래 워크시트에서 단가표[A10:D13]를 이용하여 단가[C2:C7]를 배열수식으로 계산하고자 한다. 다음 중 [C2] 셀에 입력된 수식으로 옳은 것은?

	A	B	C	D
1	제품명	수량	단가	
2	허브차	35	2,500	
3	녹차	90	4,000	
4	허브차	15	3,000	
5	녹차	20	3,000	
6	허브차	80	3,000	
7	허브차	90	3,000	
8				
9	단가표			
10		0	30	50
11	제품명	29	49	
12	허브차	3000	2,500	3,000
13	녹차	3000	3,500	4,000

- ① {=INDEX(\$B\$12:\$D\$13, MATCH(A2, \$A\$12:\$A\$13, 0), MATCH(B2, \$B\$10:\$D\$10, 1))}
- ② {=INDEX(\$B\$12:\$D\$13, MATCH(A2, \$A\$12:\$A\$13, 1), MATCH(B2, \$B\$10:\$D\$10, 0))}
- ③ {=INDEX(MATCH(A2, \$A\$12:\$A\$13, 0), MATCH(B2, \$B\$10:\$D\$10, 1), \$B\$12:\$D\$13)}
- ④ {=INDEX(MATCH(A2, \$A\$12:\$A\$13, 1), MATCH(B2, \$B\$10:\$D\$10, 0), \$B\$12:\$D\$13)}

[해설]

{=INDEX(\$B\$12:\$D\$13, MATCH(A2, \$A\$12:\$A\$13, 0), MATCH(B2, \$B\$10:\$D\$10, 1))}

①

②

③

- ① MATCH(A2, \$A\$12:\$A\$13, 0) : [A12:A13] 영역에서 [A2] 셀, 즉 "허브차"와 동일한 값을 찾은 후 상대 위치를 표시하면 결과는 1입니다.
 - MATCH(찾을 값, 범위, 옵션) 함수에서 '옵션'을 0으로 지정하면 '찾을 값'과 정확히 일치하는 값을 찾습니다.
 - 여러 셀에 결과를 구해야 하므로 범위는 절대 참조로 지정해야 합니다.
- ② MATCH(B2, \$B\$10:\$D\$10, 1) : [B10:D10] 영역에서 [B2] 셀, 즉 35보다 작거나 같은 값 중에서 가장 근접한 값(30)을 찾은 후 상대 위치를 표시하면 결과값은 2입니다.
- ③ =INDEX(\$B\$12:\$D\$13, ①, ②) → =INDEX(\$B\$12:\$D\$13, 1, 2) : [B12:D13] 영역에서 1행 2열 즉, [C12] 셀의 값 2500을 반환합니다.

22. 셀 범위 [A1:C4]에 대한 각 보기의 수식을 실행하였을 때 다음 중 결과 값이 다른 것은?

	A	B	C
1	바나나	33	2500
2	오렌지	25	1500
3	사과	41	1200
4	배	40	2300

① =INDEX(A1:C4, MATCH("배", A1:A4, 0), 1)

② =INDEX(A1:C4, 4, 2)

③ =INDEX(A1:C4, MATCH(2300, C1:C4, 0), 2)

④ =INDEX(B3:C4, 2, 1)

[해설]

① =INDEX(A1:C4, MATCH("배", A1:A4, 0), 1)

①
②

① MATCH("배", A1:A4, 0) : [A1:A4] 영역에서 '배'가 입력된 셀의 상대 위치를 계산합니다(4).

② =INDEX(A1:C4, ①, 1) → =INDEX(A1:C4, 4, 1) : [A1:C4] 영역에서 4행 1열, 즉 [A4] 셀에 입력된 값인 배를 표시합니다.

② =INDEX(A1:C4, 4, 2) : [A1:C4] 영역에서 4행 2열, 즉 [B4] 셀에 입력된 값인 40을 출력합니다.

③ =INDEX(A1:C4, MATCH(2300, C1:C4, 0), 2)

①
②

① MATCH(2300, C1:C4, 0) : [C1:C4] 영역에서 2300이 입력된 셀의 상대 위치를 계산합니다(4).

② =INDEX(A1:C4, ①, 2) → =INDEX(A1:C4, 4, 2) : [A1:C4] 영역에서 4행 2열, 즉, [B4] 셀에 입력된 값인 40을 출력합니다.

④ =INDEX(B3:C4, 2, 1) : [B3:C4] 영역에서 2행 1열, 즉 [B4] 셀에 입력된 값인 40을 출력합니다.

23. 아래 시트에서 각 부서마다 직위별로 종합점수의 합계를 구하려고 한다. 다음 중 [B17] 셀에 입력된 수식으로 옳은 것은?

	A	B	C	D	E
1	부서명	직위	업무평가	구술평가	총점점수
2	영업부	사원	35	30	65
3	총무부	대리	38	33	71
4	총무부	과장	45	36	81
5	총무부	대리	35	40	75
6	영업부	과장	46	39	85
7	홍보부	과장	30	37	67
8	홍보부	부장	41	38	79
9	총무부	사원	33	29	62
10	영업부	대리	36	34	70
11	홍보부	대리	27	36	63
12	영업부	과장	42	39	81
13	영업부	부장	40	39	79
14					
15					
16	부서명	부장	과장	대리	
17	영업부				
18	총무부				
19	홍보부				

- ① {=SUMIFS(\$E\$2:\$E\$13, \$A\$2:\$A\$13, \$A\$17, \$B\$2:\$B\$13, \$B\$16)}
- ② {=SUM((\$A\$2:\$A\$13=A17) * (\$B\$2:\$B\$13=B16) * \$E\$2:\$E\$13)}
- ③ {=SUM((\$A\$2:\$A\$13=\$A17) * (\$B\$2:\$B\$13=B16) * \$E\$2:\$E\$13)}
- ④ {=SUM((\$A\$2:\$A\$13=A17) * (\$B\$2:\$B\$13=\$B16) * \$E\$2:\$E\$13)}

[해설]

- 합계를 구하는 배열 수식은 다음의 두 가지 식이 있으며, 조건의 개수에 따라 조건을 지정하는 부분만 늘어납니다.

- 방법1 : {=SUM((조건1)*(조건2)*합계를_구할_범위) }
- 방법2 : {=SUM(IF((조건1)*(조건2), 합계를_구할_범위)) }

1. 조건과 범위 찾기

- 조건1 : 부서마다란 조건은, 비교 대상이 될 부서명 범위 [A2:A13]과 비교할 기준이 되는 “영업부”가 들어있는 [A17] 셀을 “=”으로 연결하여 적어주면 됩니다(A2:A13=A17).
- 조건2 : 직위별이란 조건은, 비교 대상이 될 직위 범위 [B2:B13]과 비교할 기준이 되는 “부장”이 들어있는 [B16] 셀을 “=”으로 연결하여 적어주면 됩니다(B2:B13=B16).
- 합계를_구할_범위 : 총점점수이므로 [E2:E13]이 됩니다.

2. 위의 조건과 범위를 합계 구하기 배열 수식에 대입하면 다음과 같습니다

- 방법1 : {=SUM((A2:A13= A17) * (B2:B13= B16) * E2:E13) }
- 방법2 : {=SUM(IF((A2:A13= A17) * (B2:B13= B16), E2:E13)) }

- 이 문제는 여러 셀에 결과값을 구하는 수식으로, 범위는 절대 참조로 지정해야 하지만, A17 셀의 경우는 A18, A19와 같이 열을 고정되고 행만 변경되어야 하므로 \$A17로 지정하고, B16 셀의 경우는 C16, D16과 같이 행은 고정되고 열만 변경되어야 하므로 B\$16으로 지정해야 합니다.
- ‘방법1’을 =SUM((\$A\$2:\$A\$13=\$A17)*(\$B\$2:\$B\$13=B\$16)*\$E\$2:\$E\$13)으로 입력한 후 [Ctrl]+[Shift]+[Enter]를 누르면 중괄호({ })가 표시되어 {=SUM((\$A\$2:\$A\$13=\$A17)*(\$B\$2:\$B\$13=B\$16)*\$E\$2:\$E\$13)}으로 표시됩니다.

24. 다음과 같이 [A1:A6]의 이름이 SCORES일 때 [A7] 셀에 아래의 함수를 입력하였다. 그 결과 값으로 옳지 않은 것은?

	A
1	2
2	2
3	0
4	1
5	TRUE
6	사용불가

- ① =ROUNDUP(AVERAGE(SCORES), 0) → 2
 ② =TRUNC(SUM(SCORES)/COUNT(SCORES), 0) → 2
 ③ =ROUND(SUM(SCORES)/COUNTA(SCORES), 0) → 1
 ④ =AVERAGEA(A1:A6) → 1

[해설]

① =ROUNDUP(AVERAGE(SCORES), 0)

①
②

① AVERAGE(SCORES) : SCORES로 이름 정의된 영역(A1:A6)의 평균인 $5/4=1.25$ 를 반환합니다.

※ 논리값 TRUE가 숫자로 처리되지 않아 계산 시 제외됩니다.

② =ROUNDUP(①, 0) → =ROUNDUP(1.25, 0) : 1.25를 소수점 이하 첫째 자리에서 올림하여 2를 반환합니다.

② =TRUNC(SUM(SCORES) / COUNT(SCORES), 0)

①
②
③

① SUM(SCORES) : SCORES로 이름 정의된 영역(A1:A6)의 합계인 5를 반환합니다.

② COUNT(SCORES) : SCORES로 이름 정의된 영역(A1:A6)에서 숫자가 들어 있는 셀의 개수인 4를 반환합니다.

③ =TRUNC(①/②, 0) → =TRUNC(5/4, 0) : 5를 4로 나눈 후 소수점 이하를 버린 1을 반환합니다.

③ =ROUND(SUM(SCORES) / COUNTA(SCORES), 0)

①
②
③

① SUM(SCORES) : SCORES로 이름 정의된 영역(A1:A6)의 합계인 5를 반환합니다.

② COUNTA(SCORES) : SCORES로 이름 정의된 영역(A1:A6)에서 데이터가 들어 있는 셀의 개수인 6을 반환합니다.

③ =ROUND(①/②, 0) → =ROUND(5/6, 0) : 5를 6으로 나눈 후 소수점 이하 첫째 자리에서 반올림하여 1을 반환합니다.

④ =AVERAGEA(A1:A6) : 수치가 아닌 셀을 포함하는 인수의 평균인 $6/6=1$ 을 반환합니다.

※ 논리값 TRUE가 숫자 1로 처리되어 계산 시 포함됩니다.

[엑세스 함수]

25. 다음과 같은 식을 입력하였을 때의 설명으로 틀린 것은?

=Format(Now(), "m/d")

- ① Format은 계산 결과에 표시 형식을 지정하는 함수이다.
- ② Now는 현재 날짜와 시간을 표시해주는 함수이다.
- ③ 컨트롤에 입력되는 식은 =로 시작해야 한다.
- ④ 오늘 날짜가 '2024-06-03'이면 06/03으로 표시된다.

[해설]

Format 함수의 표시 형식으로 m/d와 같이 월과 일이 모두 한 자리로 지정되었으므로 오늘 날짜가 2024-06-03인 경우 6/3으로 표시됩니다.

26. [매출 실적 관리] 품의 'txt평가' 컨트롤에는 'txt매출수량' 컨트롤의 값이 1,000 이상이면 "우수", 500 이상이면 "보통", 그 미만이면 "저조"라고 표시하고자 한다. 다음 중 'txt평가'의 컨트롤 원본으로 옳지 않은 것은?

- ① =IIf([txt매출수량]<500, "저조", IIf([txt매출수량]>=1000, "우수", "보통"))
- ② =IIf([txt매출수량]<500, "저조", IIf([txt매출수량]>=500, "보통", "우수"))
- ③ =IIf([txt매출수량]>=1000, "우수", IIf([txt매출수량]>=500, "보통", "저조"))
- ④ =IIf([txt매출수량]>=500, IIf([txt매출수량]<1000, "보통", "우수"), "저조")

[해설]

- ②번 함수식은 [txt매출수량]이 500 미만이면 "저조", 500 이상이면 "보통", 나머지는 "우수"로 표시합니다.
- [txt매출수량]이 1000 이상이어도 500 이상일 경우에 해당하므로 "우수"가 아닌 "보통"이 표시됩니다.

27. 다음 중 보고서에서 '페이지 번호'를 표현하는 식을 다음과 같이 설정한 경우 페이지 번호가 표시되는 결과로 옳은 것은?

=IIf([page] Mod 2 = 0, [page] & "페이지", " ")

- ① 짝수 페이지에는 "2"와 같이 표시되고 홀수 페이지에는 표시되지 않는다.
- ② 홀수 페이지에는 "1"과 같이 표시되고 짝수 페이지에는 표시되지 않는다.
- ③ 짝수 페이지에는 "2페이지"와 같이 표시되고 홀수 페이지에는 표시되지 않는다.
- ④ 홀수 페이지에는 "1페이지"와 같이 표시되고 짝수 페이지에는 표시되지 않는다.

[해설]

=IIf([page] Mod 2 = 0, [page] & "페이지", " ")

①

②

- ① [page] Mod 2 : 현재 페이지 번호를 2로 나눈 나머지를 반환합니다.
- ② IIf(①=0, [page] & "페이지", " ") : ①이 0이면, 즉 현재 페이지가 짝수 페이지면, 현재 페이지 번호에 "페이지"를 붙여 표시하고 그렇지 않고 홀수 페이지면 아무것도 표시하지 않습니다.
- ※ 현재 페이지 번호가 2로 짝수 페이지면 2페이지와 같이 현재 페이지에 "페이지"를 붙여 표시하고 현재 페이지 번호가 1로 홀수 페이지라면 아무것도 표시하지 않습니다.

28. 다음 중 폼 바닥글에 있는 텍스트 상자의 컨트롤 원본에 <사원> 테이블에서 직급이 '부장'인 레코드들의 급여 평균을 구하는 함수식으로 옳은 것은?

- ① =DAVG("[급여]", "[사원]", "[직급]='부장'")
- ② =DAVG("[사원]", "[급여]", "[직급]='부장'")
- ③ =AVG("[급여]", "[사원]", "[직급]='부장'")
- ④ =AVG("[사원]", "[급여]", "[직급]='부장'")

[해설]

=DAVG("[급여]", "[사원]", "[직급]='부장'")의 의미

- 급여 : 결과 값을 구할 필드 이름으로, 여기서는 급여의 평균을 구하므로 '급여' 필드를 지정합니다.
- 사원 : 작업 대상 레코드가 들어 있는 테이블이나 쿼리의 이름을 지정합니다.
- 직급=부장 : 조건으로, '직급'이 "부장"인 레코드를 대상으로 합니다.

29. 도서를 관리하기 위한 다음 폼은 <도서> 테이블을 레코드 원본으로 사용한다. <도서> 테이블에는 텍스트 형식의 '저자코드' 필드가 있으며, '저자명' 필드는 <저자> 테이블에 존재한다. '저자코드'를 표시하는 'txt저자코드' 컨트롤을 이용하여 'txt저자명' 컨트롤에 '저자명'을 표시하도록 하기 위한 컨트롤 원본으로 가장 적절한 것은?

도서(도서코드, 저자코드, 출판사, 도서명, 가격)
저자(저자코드, 저자명)

도서코드	저자코드	저자명
1011	103	김용갑
1012	104	김우경
1013	105	김종일

- ① =DLookup(저자명, 저자, "저자코드=" & [txt저자코드] & "")
- ② =DLookup("저자명", "저자", "저자코드=" & [txt저자코드] & "")
- ③ =DLookup("저자", "저자명", "저자코드=" & [txt저자코드] & "")
- ④ =DLookup(저자, 저자명, "저자코드=" & [txt저자코드] & "")

[해설]

=DLookup("저자명", "저자", "저자코드=" & [txt저자코드] & "")의 의미

- 도메인 함수에 사용되는 인수들은 각각을 큰따옴표(" ")로 묶어야 합니다.
- 저자명 : 결과 값을 구할 필드 이름으로, 여기서는 저자명을 표시하므로 '저자명' 필드를 지정합니다.
- 저자 : 작업 대상 레코드가 들어 있는 테이블이나 쿼리의 이름을 지정합니다.
- 저자코드='txt저자코드' : 조건으로, '저자코드'가 'txt저자코드' 컨트롤에 입력된 값과 같은 자료를 대상으로 합니다. 단 'txt저자코드' 컨트롤에 입력된 값이 텍스트일 경우에는 & 연사자를 이용하여 'txt저자코드' 컨트롤의 앞과 뒤를 작은따옴표('')와 연결해야 합니다.

30. 다음 중 [학생] 테이블에서 '학년' 필드가 1인 레코드의 개수를 계산하고자 할 때의 수식으로 옳은 것은? (단, [학생] 테이블의 기본 키는 '학번' 필드이다.)

- ① =DLookup(" ", "학생", "학년=1")
- ② =DLookup(*, 학생, 학년=1)
- ③ =DCount(학번, 학생, 학년=1)
- ④ =DCount(" ", "학생", "학년=1")

[해설]

=DCount(" ", "학생", "학년=1")의 의미

- 도메인 함수에 사용되는 인수들은 각각을 큰따옴표(" ")로 묶어야 합니다.
- * : 결과 값을 구할 필드 이름으로, *는 모든 레코드를 표시하는 만능 문자입니다.
- 학생 : 작업 대상 레코드가 들어 있는 테이블이나 쿼리의 이름을 지정합니다.
- 학년=1 : 조건으로, 1학년인 레코드를 대상으로 합니다.



시나공